# $Id: PKGBUILD 171546 2012-11-19 03:43:07Z dan $
# Maintainer: Ionut Biru <ibiru@archlinux.org>
# Contributor: Gerardo Exequiel Pozzi <vmlinuz386@yahoo.com.ar>

pkgname=libcap-ng
pkgver=0.7.3
pkgrel=1
pkgdesc="A library making programming with POSIX capabilities easier than traditional libcap"
arch=('i686' 'x86_64' sh4)
url="http://people.redhat.com/sgrubb/libcap-ng/"
license=('GPL2' 'LGPL2.1')
depends=('glibc')
options=('!libtool')
source=(http://people.redhat.com/sgrubb/$pkgname/$pkgname-$pkgver.tar.gz)
md5sums=('610afb774f80a8032b711281df126283')

build() {
  cd $pkgname-$pkgver

  ./configure --prefix=/usr --enable-static=no --with-python=no
  sed -i 's/fcntl.h/asm-generic\/fcntl.h/' utils/pscap.c
  sed -i 's/fcntl.h/asm-generic\/fcntl.h/' utils/netcap.c
  sed -i 's/fcntl.h/asm-generic\/fcntl.h/' utils/filecap.c
  make
}

package() {
  cd $pkgname-$pkgver

  make DESTDIR=$pkgdir install
}

# vim:set ts=2 sw=2 et:
